HEWLETT-PACKARD

HP-41C

USERS' LIBRARY SOLUTIONS

Optometry II (Contact Lens)



NOTICE

The program material contained herein is supplied without representation or warranty of any kind. Hewlett-Packard Company therefore assumes no responsibility and shall have no liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof.

INTRODUCTION

This HP-41C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become and expert on your HP calculator.

KEYING A PROGRAM INTO THE HP-41C

There are several things that you should keep in mind while you are keying in programs from the program listings provided in this book. The output from the HP 82143A printer provides a convenient way of listing and an easily understood method of keying in programs without showing every keystroke. This type of output is what appears in this handbook. Once you understand the procedure for keying programs in from the printed listings, you will find this method simple and fast. Here is the procedure:

1. At the end of each program listing is a listing of status information required to properly execute that program. Included is the SIZE allocation required. Before you begin keying in the program, press [XEQ] [ALPHA] SIZE [ALPHA] and specify the allocation (three digits; e.g., 10 should be specified as 010).

   Also included in the status information is the display format and status of flags important to the program. To ensure proper execution, check to see that the display status of the HP-41C is set as specified and check to see that all applicable flags are set or clear as specified.

2. Set the HP-41C to PRGM mode (press the [PRGM] key) and press [■] [GTO] [•] [•] to prepare the calculator for the new program.

3. Begin keying in the program. Following is a list of hints that will help you when you key in your programs from the program listings in this handbook.

   a. When you see " (quote marks) around a character or group of characters in the program listing, those characters are ALPHA. To key them in, simply press [ALPHA], key in the characters, then press [ALPHA] again. So "SAMPLE" would be keyed in as [ALPHA] "SAMPLE" [ALPHA].

   b. The diamond in front of each LBL instruction is only a visual aid to help you locate labels in the program listings. When you key in a program, ignore the diamond.

   c. The printer indication of divide sign is /. When you see / in the program listing, press [÷].

   d. The printer indication of the multiply sign is ÷. When you see ÷ in the program listing, press [×].

   e. The ⊢ character in the program listing is an indication of the [APPEND] function. When you see ⊢, press [■] [APPEND] in ALPHA mode (press [■] and the K key).

   f. All operations requiring register addresses accept those addresses in these forms:

      nn (a two-digit number)
      IND nn (INDIRECT: [■], followed fy a two-digit number)
      X, Y, Z, T, or L (a STACK address: [•] followed by X, Y, Z, T, or L)
      IND X, Y, Z, T or L (INDIRECT stack: [■] [•] followed by X, Y, Z, T, or L)

      Indirect addresses are specified by pressing [■] and then the indirect address. Stack addresses are specified by pressing [•] followed by X, Y, Z, T, or L. Indirect stack addresses are specified by pressing [■] [•] and X, Y, Z, T, or L.

Printer Listing	Keystrokes	Display
01◆LBL "SAM PLE"	[■] [LBL] [ALPHA] SAMPLE [ALPHA]	*01 LBL"SAMPLE*
02 "THIS IS A "	[ALPHA] *THIS IS A* [ALPHA]	*02"THIS IS A*
03 "⊢SAMPLE "	[ALPHA] [■] [APPEND] SAMPLE	*03" ⊢ SAMPLE*
04 AVIEW	[■] AVIEW [ALPHA]	*04 AVIEW*
05 6	6	*05 6*
06 ENTER↑	[ENTER↑]	*06 ENTER ↑*
07 -2	2 [CHS]	*07 -2*
08 /	[÷]	*08 /*
09 ABS	[XEQ] [ALPHA] ABS [ALPHA]	*09 ABS*
10 STO IND L	[STO] [■] [•] L	*10 STO IND L*
11 "R3="	[ALPHA] R3= [■] [ARCL] 03	*11"R3=*
12 ARCL 03	[■] [AVIEW]	*12 ARCL 03*
13 AVIEW	[ALPHA]	*13 AVIEW*
14 RTN	[■] [RTN]	*14 RTN*

TABLE OF CONTENTS

* Requires one memory module.

BACK VERTEX POWER OF PMMA CONTACT LENS

This program calculates the front and back surface powers and back vertex power of a PMMA contact lens. The inputs are the front and back radii of curvature and the center thickness.

Equations:

$$FSP = \frac{(n'-n)\ 1000}{FSR}$$

$$BSP = \frac{(n'-n)\ 1000}{-BSR}$$

$$BVP = \frac{F_1 + F_2 - \left(\frac{T}{1000\ n'}\right) F_1 F_2}{1 - \left(\frac{T}{1000\ n'}\right) F_1}$$

where:

FSP = Front surface power in diopters

BSP = Back surface power in diopters

BVP = Back vertex power in diopters

FSR = Front surface radius in mm

BSR = Back surface radius in mm

T = Center thickness in mm

n = Index of refraction for air (1.00)

n' = Index of refraction for PMMA (1.49)

Reference:

J. R. Roggenkamp
Pacific University College of Optometry
Forest Grove, OR

Example:

What are the powers of the front and back surfaces and back vertex power of a contact lens of PMMA with the following specifications:

FSR = 8.00mm

BSR = 7.40mm

T = .15mm

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 006	
[XEQ] [ALPHA] BVP [ALPHA]	FSR ?
8 [R/S]	BSR ?
7.4 [R/S]	T ?
.15 [R/S]	FSP=61.25
[R/S]	BSP=-66.22
[R/S]	BVP=-4.59

User Instructions

SIZE: 006

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize.		[XEQ] BVP	FSR ?
3.	Key in the front surface radius.	FSR (mm)	[R/S]	BSR ?
4.	Key in the back surface radius.	BSR (mm)	[R/S]	T ?
5.	Key in the center thickness.	T (mm)	[R/S]	FSP=
			[R/S]	BSP=
			[R/S]	BVP=
	(Outputs are in diopters.)			

Program Listings

```
01♦LBL "BVP
"
02 "FSR ?"
03 PROMPT
04 STO 01
05 "BSR ?"
06 PROMPT
07 STO 00
08 "T ?"
09 PROMPT
10 1490
11 /
12 STO 02
13 490
14 RCL 01
15 /
16 STO 04
17 "FSP="
18 ARCL X
19 AVIEW
20 STOP
21 490
22 RCL 00
23 /
24 CHS
25 STO 03
26 "BSP="
27 ARCL X
28 AVIEW
29 STOP
30 RCL 04
31 *
32 RCL 02
33 *
34 STO 05
35 RCL 04
36 RCL 03
37 +
38 RCL 05
39 -
40 STO 05
41 RCL 02
42 RCL 04
43 *
44 1
45 X<>Y
46 -
47 RCL 05
48 X<>Y
49 /
50 "BVP="
51 ARCL X
52 AVIEW
53 RTN
54 .END.
```

Input

T/1000n'

Calculate FSP

Calculate BSP

Calculate BVP

60 70 80 90 00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	BSR	50	
	FSR		
	T/1000n'		
	BSP		
	FSP		
05	Scratch	55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 006 TOT. REG. 19 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

EFFECTIVE POWER OF SPECTACLE LENSES AT CORNEAL PLANE

This program calculates the power of a contact lens needed for a given spectacle prescription or vice versa at a given vertex distance.

Equations:

$$F_c = \frac{F_s}{1 - \frac{d\ F_s}{1000}} \qquad F_s = \frac{F_c}{1 + \frac{d\ F_c}{1000}}$$

where:

F_c = Power at cornea in diopters.

F_s = Power at spectacle plane in diopters.

d = Distance at which spectacle lens is worn in mm.

Reference:

J. R. Roggenkamp
Pacific University College of Optometry
Forest Grove, OR

Example:

What power contact lens is needed to correct an ametropia which is correctable by a -9.00 spectacle lens worn at 13mm?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 002	
[XEQ] [ALPHA] EFF [ALPHA]	FC ?
[R/S]	FS ?
9 [CHS] [R/S]	d ?
13 [R/S]	FC=-8.06

User Instructions

SIZE: 002

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize.		[XEQ] EFF	FC ?
3.	Key in the power at cornea if known,			
	otherwise, just press [R/S].	F_c (D.)	[R/S]	FS ?
4.	Key in the power at spectacle plane, if			
	known, otheriwse, just press [R/S]	F_s (D.)	[R/S]	d ?
5.	Key in the distance at which the spectacle			
	lens is worn.	d (mm)	[R/S]	FC=
				or FS=

Program Listings

```
01◆LBL "EFF
"
02 CF 22
03 "FC ?"
04 PROMPT
05 STO 00
06 CF 00
07 FC?C 22
08 SF 00
09 "FS ?"
10 PROMPT
11 FS? 00
12 STO 00
13 "d ?"
14 PROMPT
15 1000
16 /
17 STO 01
18 RCL 00
19 1
20 RCL 01
21 RCL 00
22 *
23 FS? 00
24 CHS
25 +
26 /
27 "FS="
28 FS? 00
29 "FC="
30 ARCL X
31 AVIEW
32 RTN
33 .END.
```

Input

Is F_c unknown?

Calculate F_c or F_s

Output

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	F_c or F_s	50	
	d/1000		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 002 TOT. REG. 11 USER MODE

ENG ___ FIX 2 SCI ___ ON ___ OFF ___

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		F_c unknown	F_s unknown

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

RESIDUAL CYLINDER INDUCED AT TEAR/CORNEA INTERFACE BY CONTACT LENS

This program calculates the power needed to correct the amount of residual cylinder caused by the difference in index at the tear/cornea interface when a contact lens is placed on a toric cornea.

Equations:

$$F = \frac{1000\ (n'-n)}{r_f} - \frac{1000\ (n'-n)}{r_s}$$

$$r_f = \frac{337.5}{K_f/1000} \qquad r_s = \frac{337.5}{K_s/1000}$$

where:

F = Correcting power in diopters

K_f = Flattest K in diopters

K_s = Steepest K in diopters

r_f = Radius of cornea at K_f

r_s = Radius of cornea at K_s

n' = Cornea index of refraction (1.376)

n = Tear index of refraction (1.336)

Reference:

J. R. Roggenkamp
Pacific University College of Optometry
Forest Grove, OR

Example:

What amount of correction would be needed to correct the cylinder induced at the tear/cornea interface when placing a contact lens on a cornea with the following K readings:

K_s = 48.0 @ 90°

K_f = 42.0 @ 180°

Keystrokes:	Display:	
[XEQ] [ALPHA] SIZE [ALPHA] 001		
[XEQ] [ALPHA] TEAR [ALPHA]	STEEPEST K?	
48 [R/S]	FLATTEST K?	
42 [R/S]	F=-0.71	(x180°)

User Instructions

SIZE: 001

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize.		[XEQ] TEAR	STEEPEST K?
3.	Key in K_s.	K_s (D.)	[R/S]	FLATTEST K?
4.	Key in K_f.	K_f (D.)	[R/S]	F=

Program Listings

```
 01◆LBL "TEA
R"
 02 "STEEPES
T K?"
 03 PROMPT
 04 STO 00
 05 "FLATTES
T K?"
 06 PROMPT
 07 337.5
 08 X<>Y
 09 /
 10 40
 11 X<>Y
 12 /
 13 337.5
 14 RCL 00
 15 /
 16 40
 17 X<>Y
 18 /
 19 -
 20 "F="
 21 ARCL X
 22 AVIEW
 23 RTN
 24 .END.
```

Input

Calculation

Output

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	K_S	50	
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 001 TOT. REG. 10 USER MODE
ENG ____ FIX 2 SCI ____ ON ____ OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

CYLINDER INDUCED BY TORIC CONTACT LENS

This program calculates the cylindrical correction required to correct the residual astigmatism caused by the difference between the indices of the posterior surface of the contact lens and the tear layer.

The index of refraction for tears is taken as 1.336. For plastic, n is 1.49.

Note that the contact lens back surface parameters can be entered in either mm or diopters.

Reference:

J. R. Roggenkamp
Pacific University College of Optometry
Forest Grove, OR

Example:

What is the amount of cylinder needed to correct the cylinder caused by a toric base curve contact lens of 49.0 D at 180° and 45.0 D at 90°?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 005	
[XEQ] [ALPHA] TORIC [ALPHA]	STEEPEST K?
49 [R/S]	FLATTEST K?
45 [R/S]	FLAT K MER?
90 [R/S]	F=-1.83
[R/S]	180.00 DEG

User Instructions

SIZE : 005

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize.		[XEQ] TORIC	STEEPEST K?
3.	Key in steepest K.	K_s (D or mm)	[R/S]	FLATTEST K?
4.	Key in flattest K.	K_f (D or mm)	[R/S]	FLAT K MER?
5.	Key in meridian of flat K.	mer (deg)	[R/S]	F=
			[R/S]	( ) DEG

Program Listings

```
 01*LBL "TOR
IC"
 02 "STEEPES
T K?"
 03 PROMPT
 04 STO 01
 05 20
 06 X<>Y
 07 X<=Y?
 08 GTO 00
 09 337.5
 10 RCL 01
 11 /
 12*LBL 00
 13 STO 04
 14 "FLATTES
T K?"
 15 PROMPT
 16 STO 02
 17 20
 18 X<>Y
 19 X<=Y?
 20 GTO 00
 21 337.5
 22 RCL 02
 23 /
 24*LBL 00
 25 STO 00
 26 "FLAT K
MER?"
 27 PROMPT
 28 STO 03
 29 -154
 30 RCL 04
 31 /
 32 -154
 33 RCL 00
 34 /
 35 X<>Y
 36 -
 37 CHS
 38 "F="
 39 ARCL X
 40 AVIEW
 41 STOP
 42 RCL 03
 43 90
 44 X<Y?
 45 CHS
 46 +
 47 CLA
```

Input K_s and convert to mm if necessary

Input K_f and convert to mm if necessary

Input K_f meridian

Calculate F

Calculate axis

```
 48 ARCL X
 49 "⊢ DEG"
 50 AVIEW
 51 RTN
 52 .END.
```

60	
70	
80	
90	
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	K_f (mm)	50	
	K_s (D.)		
	K_f (D.)		
	meridian		
	K_s (mm)		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 005 TOT. REG. 22 USER MODE
ENG ____ FIX 2 SCI ____ ON ____ OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

CONTACT LENS POWER NECESSARY TO CORRECT AMETROPIA

This program calculates the power (in minus cylinder form) of a contact lens needed to fully correct a patient's ametropia.

Notes:

The spectacle R_x is to be entered in minus cylinder form.

The contact lens is a sphere on the back surface.

The correct sphere to order is the spherical equivalent of the answer given.

Reference:

J. R. Roggenkamp
Pacific University College of Optometry
Forest Grove, OR

Example:

What power back surface sphere would need to be ordered to correct the ametropia of a patient with a spectacle R_x of -2.0 -3.0x180 and K readings of 47.0 and 45.0, if he is to be fit with a 45.5 D contact lens inner surface?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 007	
[XEQ] [ALPHA] AMETROP [ALPHA]	SPH↑ CYL↑ AX
2 [CHS] [ENTER↑]	
3 [CHS] [ENTER↑]	
180 [R/S]	FLATTEST K?
45 [R/S]	STEEPEST K?
47 [R/S]	BACK K?
45.5 [R/S]	SPH=-2.50
[R/S]	CYL=-1.00
[R/S]	AX=180

User Instructions

SIZE: 007

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize.		[XEQ] AMETROP	SPH↑ CYL↑ AX
3.	Key in sphere of spectacle R_x.	SPH (D.)	[ENTER↑]	
4.	Key in cylinder of spectacle R_x.	CYL (D.)	[ENTER↑]	
5.	Key in axis of spectacle R_x.	AX (Deg)	[R/S]	FLATTEST K?
6.	Key in flattest K of cornea.	K_f (D.)	[R/S]	STEEPEST K?
7.	Key in steepest K of cornea.	K_s (D.)	[R/S]	BACK K?
8.	Key in back K of contact lens.	K_b (D.)	[R/S]	SPH=
			[R/S]	CYL=
			[R/S]	AX=

Program Listings

```
 01◆LBL "AME
TROP"
 02 "SPH↑ CY
L↑ AX"
 03 PROMPT
 04 STO 03
 05 RDN
 06 STO 02
 07 RDN
 08 STO 01
 09 "FLATTES
T K?"
 10 PROMPT
 11 STO 04
 12 "STEEPES
T K?"
 13 PROMPT
 14 STO 05
 15 "BACK K?
"
 16 PROMPT
 17 STO 06
 18 RCL 04
 19 -
 20 CHS
 21 STO 04
 22 RCL 05
 23 RCL 06
 24 -
 25 STO 05
 26 RCL 01
 27 ST+ 04
 28 RCL 02
 29 +
 30 +
 31 STO 05
 32 RCL 04
 33 -
 34 STO 00
 35 X<0?
 36 GTO 00
 37 2
 38 /
 39 ST+ 04
 40 RCL 00
 41 CHS
 42 STO 00
 43 RCL 03
 44 90
 45 X<Y?
 46 CHS
 47 +
 48 STO 03
 49◆LBL 00
 50 "SPH="
 51 ARCL 04
 52 AVIEW
 53 STOP
 54 "CYL="
 55 ARCL 00
 56 AVIEW
 57 STOP
 58 "AX="
 59 ARCL 03
 60 AVIEW
 61 RTN
 62 .END.
```

Input

Calculation

Conversion to minus cylinder form

Output

70

80

90

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Used	50	
	SPH		
	CYL		
	AX		
	K_f		
05	K_s	55	
	K_b		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 007 TOT. REG. 25 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

TORIC CONTACT LENS PARAMETERS

This program calculates the radii of curvature of the back surface and the powers at the principal meridians of a toric back surface contact lens. All keratometer entries must be in diopters. Spectacle R_x must be entered in minus cylinder form.

This program assumes that the axis of the correcting lens will correspond to the keratometer axis of the cornea. It is also assumed that the steeper corneal meridian has the most refractive power.

Reference:

J. R. Roggenkamp
Pacific University College of Optometry
Forest Grove, OR

Example:

What is the radius of curvature and power along each major meridian of a back surface toric contact lens if:

Spectacle R_x = -2.0 -3.0x180°
Steepest K = 48.0 D
Flattest K = 44.0 D

and you desire to fit 0.5 D flatter than the steepest K and 0.5 D steeper than the flattest K? (i.e., 44.5 D and 47.5 D)

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 008	
[XEQ] [ALPHA] PARA [ALPHA]	SPH↑ CYL↑ AX
2 [CHS] [ENTER↑]	
3 [CHS] [ENTER↑]	
180 [R/S]	KFR ?
44 [R/S]	KSR ?
48 [R/S]	KFCL ?
44.5 [R/S]	KSCL ?
47.5 [R/S]	P=-2.50
[R/S]	R=7.58
[R/S]	AX=180.00
[R/s]	P=-4.50
[R/S]	R=7.11
[R/S]	AX=90.00

User Instructions

SIZE: 008

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program			
2.	Initialize.		[XEQ] PARA	SPH↑ CYL↑ AX
3.	Key in sphere of R_x.	Sph (D)	[ENTER↑]	
4.	Key in cylinder of R_x.	Cyl (D)	[ENTER↑]	
5.	Key in axis of R_x.	Ax (Deg)	[R/S]	KFR ?
6.	Key in flattest K reading.	K_{FR} (D)	[R/S]	KSR ?
7.	Key in steepest K reading.	K_{SR} (D)	[R/S]	KFCL ?
8.	Key in flattest K of contact lens.	K_{FCL} (D)	[R/S]	KSCL ?
9.	Key in steepest K of contact lens.	K_{SCL} (D)	[R/S]	P= (D)
			[R/S]	R= (mm)
			[R/S]	AX= (deg)
			[R/S]	P= (D)
			[R/S]	R= (mm)
			[R/S]	AX= (deg)

Program Listings

```
 01♦LBL "PAR
A"
 02 "SPH↑ CY
L↑ AX"
 03 PROMPT
 04 STO 02
 05 RDN
 06 STO 01
 07 RDN
 08 STO 00
 09 "KFR ?"
 10 PROMPT
 11 STO 03
 12 "KSR ?"
 13 PROMPT
 14 STO 04
 15 "KFCL ?"
 16 PROMPT
 17 STO 05
 18 "KSCL ?"
 19 PROMPT
 20 STO 06
 21 RCL 03
 22 RCL 05
 23 -
 24 RCL 04
 25 RCL 06
 26 -
 27 STO 07
 28 RDN
 29 RCL 00
 30 +
 31 "P="
 32 ARCL X
 33 AVIEW
 34 STOP
 35 337.5
 36 RCL 05
 37 /
 38 "R="
 39 ARCL X
 40 AVIEW
 41 STOP
 42 "AX="
 43 ARCL 02
 44 AVIEW
 45 STOP
 46 RCL 00
 47 RCL 01
 48 +
 49 RCL 07
 50 +
 51 "P="
 52 ARCL X
 53 AVIEW
 54 STOP
 55 337.5
 56 RCL 06
 57 /
 58 "R="
 59 ARCL X
 60 AVIEW
 61 STOP
 62 RCL 02
 63 90
 64 X<Y?
 65 CHS
 66 +
 67 "AX="
 68 ARCL X
 69 AVIEW
 70 RTN
 71 .END.
```

Input

Calculate and output for flattest meridian

Calculate and output for steepest meridian

80	
90	
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	SPH	50	
	CYL		
	AX		
	K_{FR}		
	K_{SR}		
05	K_{FCL}	55	
	K_{SCL}		
	Used		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 008 TOT. REG. 28 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

TABB CONTACT LENS OF 1ST APPROXIMATION

This program calculates the parameters of a contact lens of 1st approximation as described by Roger Tabb. By entering the flattest corneal K, amount of corneal cylinder, the diameter (in the number of mm above the base curve) and the % tear reservoir desired, the calculator calculates the base curve in diopters and millimeters, overall diameter, optic zone diameter, and the peripheral curves (3 of them) radii and widths. This program may also be used to generate orthokeratology lenses for Tabb's system of orthokeratology.

Equations used:

$$\text{size} = K_f + 1.0 \text{ mm}$$

$$\text{OZR} = K_f$$

$$\text{OZD} = 2\sqrt{X\left(\frac{TD}{2}\right)^2}$$ where X = 1 - tear reservoir, TD = total diameter

$$I_1\text{CR} = \text{OZR} + 1.0 \text{ mm}$$

$$I_2\text{CR} = \text{OZR} + 2.0 \text{ mm}$$

$$I_3\text{CR} = \text{OZR} + 3.0 \text{ mm}$$

Reference:

Roger Tabb, O.D.
Westgate Square Medical Center, No. 175
3800 S.W. Cedar Hills Blvd.
Beaverton, OR

Example:

A patient's flattest K is 44.0, steepest K is 45.5 (corneal cylinder = 1.5 D) and the overall diameter is to be .5 mm larger than the base curve in millimeters. The % tear layer desired is 30%. What are the parameters of the contact lens of 1st approximation as described by the Tabb fitting system?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 012	
[XEQ] [ALPHA] TABB [ALPHA]	FLATTEST K?
44 [R/S]	CORNEAL CYL.?
1.5 [R/S]	DIA?
.5 [R/S]	% TEAR?
30 [R/S]	B CURVE = 44.50
[R/S]	C. RAD=7.58
[R/S]	DIA=8.10
[R/S]	ZONE=6.68
[R/S]	ICR1=8.60
[R/S]	ICW1=0.20
[R/S]	ICR2=9.60
[R/S]	ICW2=0.30
[R/S]	ICR3=10.60
[R/S]	ICW3=0.20

User Instructions

SIZE:012

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize.		[XEQ] TABB	FLATTEST K?
3.	Input flattest K in diopters.	K_f	[R/S]	CORNEAL CYL.?
4.	Input corneal cylinder in diopters.	K_c	[R/S]	DIA?
5.	Input number of mm greater than base curve			
	desired for diameter.	Dia	[R/S]	% TEAR?
6.	Input % of tear reservoir desired and cal-			
	culate:			
	base curve in diopters	% Tear	[R/S]	B CURVE=
	base curve radius in mm		[R/S]	C. RAD=
	optic zone diameter in mm		[R/S]	DIA=
	optic zone in mm		[R/S]	ZONE=
	ICR_1 in mm		[R/S]	ICR1=
	ICW_1 in mm		[R/S]	ICW1=
	ICR_2 in mm		[R/S]	ICR2=
	ICW_2 in mm		[R/S]	ICW2=
	ICR_3 in mm		[R/S]	ICR3=
	ICW_3 in mm		[R/S]	ICW3=

Program Listings

```
01◆LBL "TAB
B"
02 FIX 2
03 SF 21
04 "FLATTES
T K?"
05 PROMPT
06 STO 01
07 "CORNEAL
CYL.?"
08 PROMPT
09 STO 07
10 "DIA ?"
11 PROMPT
12 STO 03
13 "% TEAR
?"
14 PROMPT
15 STO 05
16 RCL 07
17 X=0?
18 GTO 01
19 1
20 RCL 07
21 X<=Y?
22 GTO 01
23 1.75
24 RCL 07
25 X<=Y?
26 GTO 02
27 RCL 01
28 .75
29 +
30 GTO 08
31◆LBL 01
32 RCL 01
33 .25
34 +
35 GTO 08
36◆LBL 02
37 RCL 01
38 .5
39 +
40◆LBL 08
41 STO 06
42 "B CURVE
"
43 XEQ 00
44 .3375
45 X<>Y
46 /
```

Initialization and storage of data

Calculate and display base curve in diopters

Convert base curve to mm and display

```
47 1 E3
48 *
49 STO 02
50 "C. RAD"
51 XEQ 00
52 RCL 03
53 +
54 XEQ 05
55 STO 00
56 "DIA"
57 XEQ 00
58 2
59 /
60 X↑2
61 1 E2
62 RCL 05
63 -
64 *
65 SQRT
66 .2
67 *
68 RND
69 STO 04
70 RCL 00
71 X<>Y
72 -
73 .2
74 /
75 FRC
76 X≠0?
77 GTO 04
78 RCL 04
79◆LBL 07
80 "ZONE"
81 XEQ 00
82 +
83 XEQ 03
84 RCL 02
85 1
86 +
87 XEQ 05
88 "ICR1"
89 XEQ 00
90 RCL 09
91 XEQ 05
92 "ICW1"
93 XEQ 00
94 RCL 02
95 2
96 +
97 XEQ 05
```

Display curve radius in mm

Determine and display optic zone parameters

Calculate and display ICR1

Calculate and display ICW1

Calculate and display ICR2

Program Listings

```
 98 "ICR2"
 99 XEQ 00
100 RCL 10
101 XEQ 05
102 "ICW2"
103 XEQ 00
104 RCL 02
105 3
106 +
107 XEQ 05
108 "ICR3"
109 XEQ 00
110 RCL 11
111 XEQ 05
112 "ICW3"
113◆LBL 00
114 "⊢="
115 ARCL X
116 AVIEW
117 RTN
118◆LBL 05
119 FIX 1
120 RND
121 FIX 2
122 RTN
123◆LBL 04
124 RCL 04
125 .1
126 -
127 GTO 07
128◆LBL 03
129 RCL 00
130 RCL 04
131 -
132 2
133 /
134 STO 08
135 .3
136 1/X
137 *
138 INT
139 10
140 /
141 STO 09
142 STO 11
143 2
144 *
145 RCL 08
146 X<>Y
147 -
148 STO 10
```

Display ICW2

Display ICR3

Display ICW3

Output routine

Rounding routine

Calculate
ICW1
ICW2
ICW3

```
149 RCL 09
150 X≠Y?
151 RTN
152 .1
153 -
154 STO 09
155 X<>Y
156 .1
157 +
158 STO 10
159 RTN
160 .END.
```

70
80
90
00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Dia	50	
	Flattest K		
	CYL		
	Dia. Factor		
	Diam		
05	% Tear	55	
	Scratch		
	CYL		
	Scratch		
	Scratch		
10	Scratch	60	
	Scratch		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 012 TOT. REG. 58 USER MODE

ENG ______ FIX ______ SCI ______ ON ______ OFF ______

DEG ______ RAD ______ GRAD ______

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

MAY-GRANT CONTACT LENS OF 1ST APPROXIMATION

This program claculates the parameters of a contact lens of 1st approximation as described by May and Grant. The flattest K and power of the contact lens are required data. The information supplied by the program is: (1) Base curve in diopters, (2) Power of lens, (3) Base curve radius in mm, (4) Optic zone diameter in mm, (5) Overall diameter, (6) Intermediate curve radius, (7) Intermediate curve width, (8) Peripheral curve radius, (9) Peripheral curve width, (10) Center thickness in mm.

Reference:

Charles H. May, O.D., F.A.A.O.
Stuart C. Grant, O.D., F.A.A.O.
Donald H. Harris, O.D., F.A.A.O.
377 East Gilbert, San Bernardino, California

Example:

What are the parameters of the lens of 1st approximation as described by May-Grant fitting method for a patient with the following:

Flattest K	44.00
Contact lens power	-3.00

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 003	
[XEQ] [ALPHA] MAYGR [ALPHA]	FLATTEST K?
44 [R/S]	LENS POWER?
3 [CHS] [R/S]	B. CURVE=7.67
[R/S]	OPT. ZONE=7.67
[R/S]	DIA=9.07
[R/S]	ICR=8.67
[R/S]	ICW=0.30
[R/S]	PCR=9.17
[R/S]	PCW=0.40
[R/S]	THICK.=0,15

User Instructions

SIZE: 003

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program.			
2.	Initialize.			FLATTEST K?
3.	Input flattest K.	K_F	[R/S]	LENS POWER?
4.	Input lens power and calculate:			
	Base Curve Radius	Lens P.	[R/S]	B. CURVE=
	Optic Zone Diameter		[R/S]	OPT. ZONE=
	Overall Diameter		[R/S]	DIA=
	ICR		[R/S]	ICR=
	ICW		[R/S]	ICW=
	PCR		[R/S]	PCR=
	PCW		[R/S]	PCW=
	Thickness		[R/S]	THICK.=
	Note: All outputs are in millimeters.			

Program Listings

```
 01◆LBL "MAY
GR"
 02 SF 21
 03 FIX 2
 04 "FLATTES
T K?"
 05 PROMPT
 06 STO 01
 07 "LENS PO
WER?"
 08 PROMPT
 09 STO 02
 10 .3375
 11 RCL 01
 12 /
 13 1 E3
 14 *
 15 STO 00
 16 "B. CURV
E="
 17 ARCL X
 18 AVIEW
 19 "OPT. ZO
NE="
 20 ARCL X
 21 AVIEW
 22 1.4
 23 +
 24 "DIA="
 25 ARCL X
 26 AVIEW
 27 RCL 00
 28 1
 29 +
 30 "ICR="
 31 ARCL X
 32 AVIEW
 33 .3
 34 "ICW="
 35 ARCL X
 36 AVIEW
 37 RCL 00
 38 1.5
 39 +
 40 "PCR="
 41 ARCL X
 42 AVIEW
 43 .4
 44 "PCW="
 45 ARCL X
 46 AVIEW
```

Initialization and input of data

Calculate base curve and optical zone then display

Calculate diameter and display

Calculate ICR and display

Calculate ICW and display

Calculate PCR and display

Calculate PCW and display

```
 47 "THICK.=
"
 48 RCL 02
 49 X<0?
 50 GTO 00
 51 .02
 52 *
 53 .18
 54 +
 55 ARCL X
 56 AVIEW
 57 STOP
 58◆LBL 00
 59 .01
 60 *
 61 .18
 62 +
 63 ARCL X
 64 AVIEW
 65 RTN
 66 .END.
```

Calculate and display thickness

80	
90	
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Base Curve	50	
	Flattest K		
	Power		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 003 TOT. REG. 28 USER MODE
ENG ____ FIX 2 SCI ____ ON ____ OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

ROGGENKAMP SPECIFICATIONS FOR PRISM BALLAST

FRONT TORIC OR PRISM BALLAST CONTACT LENS

This program calculates the parameters for a lens of 1st approximation for a front toric or fused bifocal. The flattest K, amount of corneal toricity and power of the lens are required data. The information supplied by the program is: (1) base curve in diopters, (2) base curve radius in mm, (3) overall diameter, (4) optic zone diameter of a round lens, (5) optic zone diameter of a truncated lens, (6) intermediate curve radius, (7) intermediate curve width, (8) peripheral curve radius, (9) peripheral curve width, (10) amount of prism ballast (in prism diopters), (11) amount to add to standard center thickness in mm, (12) base-apex line of right eye's prism, (13) base-apex of left eye's prism.

Notes:

Fused bifocals usually require a 0.3 mm truncation along the 180° line.

The optic zone diameter is 7.7 mm for a round lens and 7.6 for a truncated lens.

In cases of anisometropia, enter the power of the more minus lens.



Reference:

J. R. Roggenkamp
Pacific University College of Optometry
Forest Grove, OR

Example:

What parameters would be required for a front surface toric contact lens as described by Roggenkamp's method of fitting for a patient with the following:

1)	Flattest K	44.00
2)	Amount of Corneal Toricity	1.50
3)	Power of Lens	-3.00

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 006	
[XEQ] [ALPHA] LENS [ALPHA]	FLATTEST K?
44 [R/S]	CORNEAL TOR.?
1.5 [R/S]	POWER?
3 [CHS] [R/S]	CURVE=44.00
[R/S]	RAD=7.67
[R/S]	DIA=9.00
[R/S]	OZDR=7.70
[R/S]	OZDT=7.60
[R/S]	ICR=8.67
[R/S]	ICW=0.45
[R/S]	PCR=11.00
[R/S]	PCW=0.20
[R/S]	DELTA=2.00
[R/S]	CT=0.17
[R/S]	A1=105.00
[R/S]	A2=75.00

User Instructions

SIZE: 006

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program.			
2.	Initialize.		[XEQ] LENS	FLATTEST K?
3.	Input flattest K.	K_f*	[R/S]	CORNEAL TOR.?
4.	Input corneal toricity.	K_c*	[R/S]	POWER?
5.	Input power of contact lens and calculate			
	base curve.	P*	[R/S]	CURVE=
	*Input units are in diopters.			
	Calculate:			
	Base Curve Radius (mm)		[R/S]	RAD=
	Overall Diameter (mm)		[R/S]	DIA=
	Optic Zone Diameter (mm) (rounded)		[R/S]	OZDR=
	Optic Zone Diameter (mm) (truncated)		[R/S]	OZDT=
	Intermediate Curve Radius (mm)		[R/S]	ICR=
	Intermediate Curve Width (mm)		[R/S]	ICW=
	Peripheral Curve Radius (mm)		[R/S]	PCR=
	Peripheral Curve Width (mm)		[R/S]	PCW=
	Prism Ballast (Δ)		[R/S]	DELTA=
	Center Thickness App. (mm)		[R/S]	CT=
	$Axis_1$°		[R/S]	A1=
	$Axis_2$°		[R/S]	A2=

Program Listings

```
 01♦LBL "LEN
S"
 02 SF 21
 03 "FLATTES
T K?"
 04 PROMPT
 05 STO 01
 06 "CORNEAL
 TOR.?"
 07 PROMPT
 08 STO 02
 09 "POWER ?
"
 10 PROMPT
 11 STO 05
 12 RCL 02
 13 1.5
 14 X<>Y
 15 X<=Y?
 16 GTO 03
 17 3.37
 18 X<>Y
 19 X<=Y?
 20 GTO 02
 21 0
 22 VIEW X
 23 STOP
 24♦LBL 02
 25 .33
 26 *
 27 RCL 01
 28 +
 29 GTO 08
 30♦LBL 03
 31 RCL 01
 32♦LBL 08
 33 "CURVE"
 34 XEQ 00
 35 STO 04
 36 .3375
 37 X<>Y
 38 /
 39 1 E3
 40 *
 41 "RAD"
 42 XEQ 00
 43 STO 03
 44 9
 45 "DIA"
 46 XEQ 00
 47 7.7
```

Input and storage of data

Determination of base curve

Diameter

```
 48 "OZDR"
 49 XEQ 00
 50 7.6
 51 "OZDT"
 52 XEQ 00
 53 RCL 03
 54 1
 55 +
 56 "ICR"
 57 XEQ 00
 58 .45
 59 "ICW"
 60 XEQ 00
 61 RCL 04
 62 43.5
 63 X<>Y
 64 "PCR"
 65 X<=Y?
 66 GTO 04
 67 11
 68 XEQ 00
 69 GTO 09
 70 4
 71♦LBL 04
 72 12.5
 73 XEQ 00
 74♦LBL 09
 75 .2
 76 "PCW"
 77 XEQ 00
 78 RCL 05
 79 1.87
 80 X<>Y
 81 "DELTA"
 82 X<=Y?
 83 GTO 07
 84 3.87
 85 X<>Y
 86 X<=Y?
 87 GTO 10
 88 6.12
 89 X<>Y
 90 X<=Y?
 91 GTO 05
 92 7.87
 93 X<>Y
 94 X<=Y?
 95 GTO 06
 96 .75
 97 XEQ 00
 98 GTO 14
```

OZD

ICR_1

ICW_1

ICR_2

ICW_2

Prism determination

Program Listings

```
 99◆LBL 06
100 1
101 XEQ 00
102 GTO 14
103◆LBL 16
104 2
105 XEQ 00
106 GTO 13
107◆LBL 15
108 2.25
109 XEQ 00
110 GTO 12
111◆LBL 10
112 1.5
113 XEQ 00
114 GTO 11
115◆LBL 05
116 1.25
117 XEQ 00
118 GTO 11
119◆LBL 07
120 -2
121 X<>Y
122 X>Y?
123 GTO 17
124 -4
125 X<>Y
126 X>Y?
127 GTO 16
128 -8
129 X<>Y
130 X>Y?
131 GTO 15
132 2.5
133 XEQ 00
134 GTO 12
135◆LBL 17
136 1.75
137 XEQ 00
138 GTO 13
139◆LBL 12
140 "CT"
141 .22
142 XEQ 00
143 GTO 01
144◆LBL 13
145 .17
146 "CT"
147 XEQ 00
148 GTO 01
149◆LBL 11
```

Thickness determination

Prism determination

Thickness determination

```
150 .13
151 XEQ 00
152 GTO 01
153◆LBL 14
154 .1
155 XEQ 00
156◆LBL 01
157 105
158 "A1"
159 XEQ 00
160 75
161 "A2"
162 XEQ 00
163 STOP
164◆LBL 00
165 "⊢="
166 ARCL X
167 AVIEW
168 RTN
169 .END.
```

80	
90	
00	

Axis determination

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Scratch	50	
	K_f		
	K_c		
	Scratch		
	Power		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 006 TOT. REG. 63 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

BRUNGART I AND II

(requires one memory module)

This program will calculate either the refractive K values and their meridians, spectacle R_x, or over refraction using the following formulas:

$$K = CL + OR + BC - SR_x$$

$$SR_x = CL - OR + BC - K$$

$$OR = SR_x + K - CL - BC$$

where:

K = Corneal Power (N=1.3375)

CL = Contact Lens Power

OR = Over Refraction

SR_x = Spectacle R_x

BC = Contact Lens Base Curve

SR_x and OR must be entered in minus cylinder form. BC and K can be entered either in diopters or millimeters. CL is in minus cylinder form. The program will work for any combination of K, CL, OR, SR_x and BC, even if all are sphero-cylinder combinations and may include obliquely crossed axes.

If the values for BC or K are 9 or greater, they are assumed to be in diopters; values less than 9 are assumed to be in millimeters.

Reference:

This program is a rearrangement of a formula published in a periodical by "Duffens Contact Lens Co.".

Example:

SR_x = -2.50 - .50 x 120°

OR = P_L - .50 x 110°

BC = 43.00

CL = -2.75 sphere

What are the refractive K values for this cornea?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 026	
[XEQ] [ALPHA] BR [ALPHA]	SRX D↑D↑DEG?
2.5 [CHS] [ENTER↑]	
.5 [CHS] [ENTER↑]	
120 [R/S]	OR D↑D↑DEG?
0 [ENTER↑]	
.5 [CHS] [ENTER↑]	
110 [R/S]	K?
[R/S]	BC?
43 [ENTER↑]	
[ENTER↑] [ENTER↑] [R/S]	CL D↑D↑DEG?
2.75 [CHS] [ENTER↑]	
0 [ENTER↑] [R/S]	42.84 DIOP
[R/S]	70. DEG
[R/S]	42.66 DIOP
[R/S]	160. DEG

Note:

43 is entered three times for BC since both meridians are 43.00. A toric lens would be entered as BD, [ENTER↑], MERID, [ENTER↑], BC, [ENTER↑], MERID, [R/S]. CL is a sphere, thus 0 is entered for its cylinder and axis.

User Instructions

SIZE: 026

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program			
2.	Initialize.		[XEQ] BR	SRX D↑D↑DEG?
3.	If SR_x is to be solved for; just press			
	[R/S].			
	Otherwise:			
	Enter SR_x in minus cylinder form.	(D)	[ENTER↑]	
		(D)	[ENTER↑]	
		(Deg)	[R/S]	OR D↑D↑DEG?
4.	If OR is to be solved for: just press			
	[R/S].			
	Otherwise:			
	Enter OR in minus cylinder form.	(D)	[ENTER↑]	
		(D)	[ENTER↑]	
		(Deg)	[R/S]	K?
5.	If K is to be solved for; just press [R/S].			
	Otherwise:			
	Enter K's in order.			
	K (Diopters or mm),	(D or mm)	[ENTER↑]	
	Meridian,	(Deg)	[ENTER↑]	
	K,	(D or mm)	[ENTER↑]	
	Meridian.	(Deg)	[R/S]	BC?
6.	Input BC in the same manner as K in step 5.	(D or mm)	[ENTER↑]	
		(Deg)	[ENTER↑]	
		(D or mm)	[ENTER↑]	
		(Deg)	[R/S]	CL D↑D↑DEG?
7.	Input CL in minus cylinder form.	(D)	[ENTER↑]	
	cont'd.			

User Instructions

SIZE: 026

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
		(D)	[ENTER↑]	
		(Deg)	[R/S]	
	If SR_x or OR is being solved for:			( ) DIOP
			[R/S]	( ) DIOP
			[R/S]	( ) DEG
	If K is being solved for:			( ) DIOP
			[R/S]	( ) DEG
			[R/S]	( ) DIOP
			[R/S]	( ) DEG

Program Listings

```
01♦LBL "BR"
02 CLRG
03 CF 22
04 FIX 2
05 ΣREG 14
06 CF 00
07 CF 01
08 CF 02
09 "SRX"
10 XEQ 00
11 FC?C 22
12 SF 00
13 STO 02
14 RDN
15 STO 01
16 RDN
17 STO 00
18 "OR"
19 XEQ 00
20 FC?C 22
21 SF 01
22 STO 05
23 RDN
24 STO 04
25 RDN
26 STO 03
27 "K ?"
28 PROMPT
29 FC?C 22
30 SF 02
31 XEQ 16
32 STO 06
33 RDN
34 STO 07
35 RDN
36 STO 08
37 "BC ?"
38 PROMPT
39 XEQ 19
40 XEQ 16
41 STO 20
42 RDN
43 STO 21
44 RDN
45 STO 22
46 "CL"
47 XEQ 00
48 STO 12
49 RDN
50 STO 11
51 RDN
```

Initialization

Input of variables

```
52 STO 10
53 FS?C 00
54 GTO "SR"
55 FS?C 01
56 GTO "OR"
57 FC?C 02
58 GTO "BR"
59 RCL 20
60 RCL 03
61 +
62 RCL 00
63 -
64 RCL 10
65 +
66 STO 13
67 RCL 12
68 2
69 *
70 RCL 11
71 STO 23
72 P-R
73 Σ+
74 RCL 22
75 2
76 *
77 RCL 21
78 ST+ 23
79 P-R
80 Σ+
81 RCL 05
82 2
83 *
84 RCL 04
85 ST+ 23
86 P-R
87 Σ+
88 RCL 02
89 2
90 *
91 RCL 01
92 ST- 23
93 CHS
94 P-R
95 Σ+
96 RCL 16
97 RCL 14
98 R-P
99 X<>Y
100 XEQ 17
101 X<>Y
102 2
```

Solve for SR_X?

Solve for OR?

Check for error

Calculate K's and meridians

Program Listings

```
103 /
104 X<>Y
105 ST- 23
106 RCL 23
107 2
108 /
109 RCL 13
110 +
111 X<>Y
112 X≠0?
113 GTO 02
114 X<>Y
115 VIEW X
116 ADV
117 VIEW X
118 CLRG
119 RTN
120◆LBL 02
121 X<>Y
122 +
123 LASTX
124 CLA
125 ARCL X
126 "⊢ DIOP"
127 AVIEW
128 X<>Y
129 RDN
130 X<>Y
131 R↑
132 X<>Y
133 90
134 X<>Y
135 +
136 LASTX
137 X<>Y
138 180
139 X<=Y?
140 XEQ 18
141 FIX 0
142 CLA
143 ARCL Z
144 "⊢ DEG"
145 AVIEW
146 ADV
147 FIX 2
148 CLA
149 ARCL T
150 "⊢ DIOP"
151 AVIEW
152 FIX 0
153 CLA
```

Output K's and meridians

```
154 ARCL Y
155 "⊢ DEG"
156 AVIEW
157 RTN
158◆LBL 18
159 -
160 RDN
161 X<>Y
162 R↑
163 X<>Y
164 RTN
165◆LBL 17
166 X<0?
167 GTO 03
168 X<>Y
169 RTN
170◆LBL 03
171 180
172 +
173 X<>Y
174 CHS
175 RTN
176◆LBL 16
177 RDN
178 X<>Y
179 RDN
180 -
181 X≠0?
182 GTO 01
183 0
184 ENTER↑
185 LASTX
186 RTN
187◆LBL 01
188 LASTX
189 RTN
190◆LBL 19
191 STO 09
192 CLX
193 9
194 X>Y?
195 GTO 05
196 CLX
197 RCL 09
198 RTN
199◆LBL 05
200 CLX
201 337.5
202 STO 25
203 X<>Y
204 /
```

Make second meridian <180°

Check for negative axis

K and BC in mm or diopters?

Change mm to diopters

Program Listings

```
205 RCL 25
206 R↑
207 /
208 RDN
209 X<>Y
210 R↑
211 RCL 09
212 RTN
213◆LBL 00
214 "⊢ D↑D↑D
EG?"
215 PROMPT
216 RTN
217◆LBL "SR"
218 RCL 03
219 RCL 06
220 -
221 RCL 20
222 +
223 RCL 10
224 +
225 STO 13
226 RCL 12
227 2
228 *
229 RCL 11
230 STO 23
231 P-R
232 Σ+
233 RCL 05
234 2
235 *
236 RCL 04
237 ST+ 23
238 P-R
239 Σ+
240 RCL 22
241 2
242 *
243 RCL 21
244 ST+ 23
245 P-R
246 Σ+
247 RCL 08
248 2
249 *
250 RCL 07
251 ST- 23
252 CHS
253 P-R
254 Σ+
```

Prompt routine

Calculate SR_x

```
255 RCL 16
256 RCL 14
257 R-P
258 XEQ 33
259 ST- 23
260 RCL 23
261 2
262 /
263 RCL 13
264 +
265 GTO 31
266◆LBL "OR"
267 RCL 00
268 RCL 06
269 +
270 RCL 20
271 -
272 RCL 10
273 -
274 STO 13
275 RCL 12
276 2
277 *
278 RCL 11
279 ST- 23
280 CHS
281 P-R
282 Σ+
283 RCL 02
284 2
285 *
286 RCL 01
287 ST+ 23
288 P-R
289 Σ+
290 RCL 08
291 2
292 *
293 RCL 07
294 ST+ 23
295 P-R
296 Σ+
297 RCL 22
298 2
299 *
300 RCL 21
301 ST- 23
302 CHS
303 P-R
304 Σ+
305 RCL 16
```

Calculate OR

Program Listings

```
306 RCL 14
307 R-P
308 XEQ 33
309 ST- 23
310 RCL 23
311 2
312 /
313 RCL 13
314 +
315◆LBL 31
316 R↑
317 X<>Y
318 ENTER↑
319 CLX
320 +
321 X<>Y
322 R↑
323 X≠0?
324 GTO 25
325 CLA
326 ARCL Z
327 "⊢ DIOP"
328 AVIEW
329 0
330 ADV
331 CLA
332 ARCL X
333 "⊢ DIOP"
334 AVIEW
335 CLA
336 ARCL X
337 "⊢ DEG"
338 AVIEW
339 RTN
340◆LBL 25
341 X>0?
342 GTO 26
343 R↑
344 R↑
345 X<>Y
346 RDN
347 GTO 27
348◆LBL 26
349 R↑
350 R↑
351 X<>Y
352 +
353 LASTX
354 CHS
355 R↑
356 90
```

Any cylinders?

Is cyl. positive?

Change to minus cylinder form

```
357 +
358 180
359 X<=Y?
360 GTO 28
361 RDN
362◆LBL 29
363 X<>Y
364 RDN
365 X<>Y
366 R↑
367 X<>Y
368 GTO 27
369◆LBL 28
370 -
371 GTO 29
372◆LBL 27
373 CLA
374 ARCL X
375 "⊢ DIOP"
376 AVIEW
377 ADV
378 CLA
379 ARCL Y
380 "⊢ DIOP"
381 AVIEW
382 FIX 0
383 CLA
384 ARCL Z
385 "⊢ DEG"
386 AVIEW
387 RTN
388◆LBL 33
389 X<>Y
390 X<0?
391 GTO 24
392 2
393 /
394 X<>Y
395 RTN
396◆LBL 24
397 180
398 +
399 2
400 /
401 X<>Y
402 CHS
403 RTN
404 .END.
```

Rearrange data

Correct axis

Output results

Is axis negative?

Change to positive axis

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	SR_x	50	
	SR_x		
	SR_x		
	OR		
	OR		
05	OR	55	
	K		
	K		
	K		
	K		
10	CL	60	
	CL		
	CL		
	USED		
	USED		
15	USED	65	
	USED		
	USED		
	USED		
	USED		
20	BC	70	
	BC		
	BC		
	SCRATCH		
	SCRATCH		
25	SCRATCH	75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 026 TOT. REG. 114 USER MODE

ENG ___ FIX 2 SCI ___ ON ___ OFF ___

DEG X RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Solve for SR_x	
01		Solve for OR	
02		Solve for K	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

HEWLETT-PACKARD

HP-41C

USERS' LIBRARY SOLUTIONS

Bar Codes

Optometry II (Contact Lens)

OPTOMETRY II

BACK VERTEX POWER OF PMMA
CONTACT LENS
PROGRAM REGISTERS NEEDED: 14

HEWLETT PACKARD
SOLUTION BOOK:
OPTOMETRY II (C-LENS)

ROW 1 (1 - 2)

ROW 2 (3 - 8)

ROW 3 (8 - 15)

ROW 4 (16 - 21)

ROW 5 (22 - 29)

ROW 6 (30 - 42)

ROW 7 (43 - 51)

ROW 8 (51 - 54)

EFFECTIVE POWER OF SPECTACLE LENSES AT CORNEAL PLANE
PROGRAM REGISTERS NEEDED: 10

HEWLETT PACKARD
SOLUTION BOOK:
OPTOMETRY II (C-LENS)

ROW 1 (1 - 3)

ROW 2 (3 - 9)

ROW 3 (9 - 15)

ROW 4 (15 - 26)

ROW 5 (27 - 31)

ROW 6 (32 - 33)

RESIDUAL CYLINDER INDUCED AT
TEAR/CORNEA INTERFACE BY CLENS
PROGRAM REGISTERS NEEDED: 10

HEWLETT PACKARD
SOLUTION BOOK:
OPTOMETRY II (C-LENS)

ROW 1 (1 - 2)

ROW 2 (2 - 5)

ROW 3 (5 - 7)

ROW 4 (7 - 14)

ROW 5 (15 - 23)

ROW 6 (24 - 24)

CYLINDER INDUCED BY TORIC
CONTACT LENS
PROGRAM REGISTERS NEEDED: 18

HEWLETT PACKARD
SOLUTION BOOK:
OPTOMETRY II (C-LENS)

ROW 1 (1 - 2)

ROW 2 (2 - 6)

ROW 3 (7 - 14)

ROW 4 (14 - 16)

ROW 5 (17 - 23)

ROW 6 (24 - 26)

ROW 7 (26 - 32)

ROW 8 (33 - 42)

ROW 9 (43 - 49)

ROW 10 (49 - 52)

CONTACT LENS POWER NECESSARY
TO CORRECT AMETROPIA
PROGRAM REGISTERS NEEDED: 19

HEWLETT PACKARD
SOLUTION BOOK:
OPTOMETRY II (C-LENS)

ROW 1 (1 - 2)

ROW 2 (2 - 4)

ROW 3 (5 - 9)

ROW 4 (9 - 12)

ROW 5 (12 - 15)

ROW 6 (15 - 27)

ROW 7 (27 - 38)

ROW 8 (39 - 49)

ROW 9 (50 - 54)

ROW 10 (54 - 61)

ROW 11 (62 - 62)

TORIC CONTACT LENS PARAMETERS

PROGRAM REGISTERS NEEDED: 21

HEWLETT PACKARD
SOLUTION BOOK:
OPTOMETRY II (C–LENS)

ROW 1 (1 – 2)

ROW 2 (2 – 7)

ROW 3 (8 – 12)

ROW 4 (12 – 17)

ROW 5 (18 – 24)

ROW 6 (25 – 34)

ROW 7 (35 – 40)

ROW 8 (41 – 49)

ROW 9 (50 – 55)

ROW 10 (56 – 64)

ROW 11 (65 – 71)

TABB CONTACT LENS OF
1ST APPROXIMATION
PROGRAM REGISTERS NEEDED: 46

HEWLETT PACKARD
SOLUTION BOOK:
OPTOMETRY II (C-LENS)

ROW 1 (1 - 4)

ROW 2 (4 - 6)

ROW 3 (7 - 7)

ROW 4 (7 - 13)

ROW 5 (13 - 18)

ROW 6 (19 - 26)

ROW 7 (27 - 34)

ROW 8 (35 - 42)

ROW 9 (42 - 45)

ROW 10 (46 - 50)

ROW 11 (51 - 56)

ROW 12 (57 - 65)

ROW 13 (66 - 76)

ROW 14 (77 - 82)

ROW 15 (83 - 88)

ROW 16 (88 - 92)

ROW 17 (93 - 98)

ROW 18 (98 - 102)

ROW 19 (103 – 108)

ROW 20 (108 – 112)

ROW 21 (113 – 121)

ROW 22 (121 – 131)

ROW 23 (132 – 142)

ROW 24 (143 – 154)

ROW 25 (155 – 160)

MAY–GRANT CONTACT LENS OF
1ST APPROXIMATION
PROGRAM REGISTERS NEEDED: 26

HEWLETT PACKARD
SOLUTION BOOK:
OPTOMETRY II (C–LENS)

ROW 1 (1 – 3)

ROW 2 (4 – 5)

ROW 3 (6 – 7)

ROW 4 (8 – 14)

ROW 5 (15 – 17)

ROW 6 (18 – 20)

ROW 7 (20 – 25)

ROW 8 (26 – 33)

ROW 9 (33 – 38)

ROW 10 (39 – 44)

ROW 11 (44 – 47)

ROW 12 (47 – 54)

ROW 13 (55 – 62)

ROW 14 (63 – 66)

ROGGENKAMP SPECS FOR PRISM
BALLAST FRONT TORIC OR C–LENS
PROGRAM REGISTERS NEEDED: 58

HEWLETT PACKARD
SOLUTION BOOK:
OPTOMETRY II (C–LENS)

ROW 1 (1 – 3)

ROW 2 (3 – 6)

ROW 3 (6 – 7)

ROW 4 (8 – 13)

ROW 5 (13 – 20)

ROW 6 (20 – 29)

ROW 7 (29 – 34)

ROW 8 (35 – 41)

ROW 9 (41 – 46)

ROW 10 (46 – 49)

ROW 11 (50 – 54)

ROW 12 (55 – 59)

ROW 13 (59 – 64)

ROW 14 (64 – 70)

ROW 15 (71 – 76)

ROW 16 (76 – 81)

ROW 17 (81 – 86)

ROW 18 (87 – 92)

ROGGENKAMP SPECS FOR PRISM
BALLAST FRONT TORIC OR C–LENS

HEWLETT PACKARD
SOLUTION BOOK:
OPTOMETRY II (C–LENS)

ROW 19 (92 – 98)

ROW 20 (99 – 105)

ROW 21 (106 – 110)

ROW 22 (111 – 116)

ROW 23 (116 – 123)

ROW 24 (123 – 131)

ROW 25 (131 – 136)

ROW 26 (136 – 141)

ROW 27 (141 – 146)

ROW 28 (146 – 151)

ROW 29 (152 – 158)

ROW 30 (158 – 162)

ROW 31 (163 – 169)

BRUNGART I AND II

PROGRAM REGISTERS NEEDED: 92

HEWLETT PACKARD
SOLUTION BOOK:
OPTOMETRY II (C–LENS)

ROW 1 (1 – 5)

ROW 2 (6 – 11)

ROW 3 (11 – 19)

ROW 4 (19 – 27)

ROW 5 (28 – 33)

ROW 6 (34 – 40)

ROW 7 (41 – 47)

ROW 8 (47 – 55)

ROW 9 (55 – 59)

ROW 10 (59 – 70)

ROW 11 (71 – 79)

ROW 12 (80 – 91)

ROW 13 (92 – 101)

ROW 14 (101 – 111)

ROW 15 (112 – 121)

ROW 16 (122 – 127)

ROW 17 (128 – 139)

ROW 18 (139 – 145)



ROW 19 (145 – 151)

ROW 20 (151 – 156)

ROW 21 (156 – 164)

ROW 22 (165 – 173)

ROW 23 (174 – 184)

ROW 24 (185 – 196)

ROW 25 (196 – 203)

ROW 26 (204 – 215)

ROW 27 (215 – 218)

ROW 28 (218 – 225)

ROW 29 (226 – 237)

ROW 30 (238 – 246)

ROW 31 (247 – 257)

ROW 32 (258 – 266)

ROW 33 (266 – 271)

ROW 34 (272 – 283)

ROW 35 (284 – 295)

ROW 36 (295 – 304)

ROW 37 (305 – 312)

ROW 38 (313 – 324)

ROW 39 (325 – 328)

ROW 40 (329 – 334)

ROW 41 (335 – 341)

ROW 42 (341 – 349)

ROW 43 (349 – 359)

ROW 44 (359 – 368)

ROW 45 (369 – 375)

ROW 46 (375 – 380)

ROW 47 (381 – 385)

ROW 48 (386 – 392)

ROW 49 (392 – 400)

ROW 50 (401 – 405)

NOTES

NOTES

NOTES


HEWLETT
PACKARD

Hewlett-Packard Software

In terms of power and flexibility, the problem-solving potential of the HP-41C programmable calculator is nearly limitless. And in order to see the practical side of this potential, HP has different types of software to help save you time and programming effort. Every one of our software solutions has been carefully selected to effectively increase your problem-solving potential. Chances are, we already have the solutions you're looking for.

Application Pacs

To increase the versatility of your HP-41C, HP has an extensive library of "Application Pacs". These programs transform your HP-41C into a specialized calculator in seconds. Included in these pacs are detailed manuals with examples, minature plug-in Application Modules, and keyboard overlays. Every Application Pac has been designed to extend the capabilities of the HP-41C.

You can choose from:

Aviation
Clinical Lab
Circuit Analysis
Financial Decisions
Mathematics
Structural Analysis
Surveying
Securities
Statistics
Stress Analysis
Games
Home Management
Machine Design
Navigation
Real Estate
Thermal and Transport Science

Users' Library

The Users' Library provides the best programs from contributors and makes them available to you. By subscribing to the HP-41C Users' Library you'll have at your fingertips literally hundreds of different programs from many different application areas.

* Users' Library Solutions Books

Hewlett-Packard offers a wide selection of Solutions Books complete with user instructions, examples, and listings. These solution books will complement our other software offerings and provide you with a valuable tool for program solutions.

You can choose from:

Business Stat/Marketing/Sales
Home Construction Estimating
Lending, Saving and Leasing
Real Estate
Small Business
Geometry
High-Level Math
Test Statistics
Antennas
Chemical Engineering
Control Systems
Electrical Engineering
Fluid Dynamics and Hydraulics
Civil Engineering
Heating, Ventilating & Air Conditioning
Mechanical Engineering
Solar Engineering
Calendars
Cardiac/Pulmonary
Chemistry
Games
Optometry I (General)
Optometry II (Contact Lens)
Physics
Surveying

* Some books require additional memory modules to accomodate all programs.

OPTOMETRY II

BACK VERTEX POWER OF PMMA CONTACT LENS
EFFECTIVE POWER OF SPECTACLE LENSES AT CORNEAL PLANE
RESIDUAL CYLINDER INDUCED AT TEAR/CORNEA INTERFACE BY CONTACT LENS
CYLINDER INDUCED BY TORIC CONTACT LENS
CONTACT LENS POWER NECESSARY TO CORRECT AMETROPIA
TORIC CONTACT LENS PARAMETERS
TABB CONTACT LENS OF 1ST APPROXIMATION
MAY-GRANT CONTACT LENS OF 1ST APPROXIMATION
ROGGENKAMP SPECIFICATIONS FOR PRISM BALLAST FRONT TORIC OR PRISM BALLAST CONTACT LENS
BRUNGART I & II

HEWLETT PACKARD

00041-90144 Printed in U.S.A.